

Using SAS® to Compare Two Estimation Methods on the Same Outcome: Example from First Trimester Pregnancy Weights

Brandy R. Sinco, MS, University of Michigan, Ann Arbor, MI
Edith Kieffer, PhD, University of Michigan, Ann Arbor, MI
Kathleen Welch, MS, University of Michigan, Ann Arbor, MI
Diana Welmerink, MPH, University of Michigan, Ann Arbor, MI

ABSTRACT

Background. Accurate pre-pregnancy weight is important for weight gain recommendations during pregnancy. Weight gain is linear during first trimester.

Objective. Use a linear mixed model (LMM), with random slope and intercept, to predict weight at the end of the first trimester (week 13), for 276 women, from a study in which many pre-pregnancy weights were self-reported and likely inaccurate. Compare the predicted weights at week 13 from the LMM to weights computed by adding a constant per week for 13 weeks.

Methods. For a sub-sample in which the weights at week 13 were known, error variances between predicted and self-reported weights were compared with a Proc Mixed random effects model and then by using Proc IML to conduct a likelihood ratio test for a variance comparison. Graphically, Proc SGPlot produced box plots and histograms to graphically display the variances between the two methods.

Next, indicators were created for weight categories (under-weight, normal, over-weight, obese) and excessive weight gain. Accuracy of categories can be compared with Proc Freq by comparing the 95% confidence intervals for Cohen's kappas. Error rates can be compared with Proc Freq by using the McNemar test to compare error rates between the projected categories by the two prediction methods. Further, Proc Logistic can be used to evaluate accuracy by comparing the areas under the ROC curves between models using predicted and self-reported pre-pregnancy weights.

Results. The likelihood ratio test, kappa confidence interval, and McNemar's test indicated that weight prediction from the LMM had lower variance and error rates.

Outline

1)_Why Pre-Pregnancy Weight Is Important in Obstetrics	Pages 1 - 2
2)_Description of Study and Participants	Pages 2 - 3
3)_Method for Imputing Missing Pre-Pregnancy Weights	Pages 3 - 5
4)_Linear Mixed Models	Pages 5 - 6
5)_Graphical Comparison of Variances (Histograms, Box Plots, Limits of Agreement)	Pages 6 - 10
6)_Testing for Variance Differences in Proc Mixed	Pages 10 - 11
7)_Comparing Classification in Proc Freq with the McNemar Statistic and Cohen's Kappa	Pages 11 - 12
8)_Logistic Regression. Comparison of Areas Under ROC Curves Between Models	Page 12
9)_Conclusions	Page 12
10)_Acknowledgements	Page 13
11)_References	Pages 13 - 14
12)_Contact Information	Page 15

1)_Why Pre-Pregnancy Weight Is Important in Obstetrics.

The amount of weight gained during pregnancy can affect the short-term and long-term health of the mother and the infant. The American College of Obstetricians and Gynecologists recommends different

amounts of pregnancy weight gain, based on the pre-pregnancy body mass index of the woman, such as 28 - 40 pounds for underweight women, 25- 35 pounds for normal-weight, 15- 25 pounds for overweight, and 11 - 20 pounds for obese women¹.

Pre-pregnancy weight is also related to risk factors during pregnancy. According to the Mayo Clinic²:

“Being overweight before pregnancy increases the risk of various pregnancy complications, including gestational diabetes, high blood pressure disorders of pregnancy, including preeclampsia, and the need for a C-section.”

An accurate pre-pregnancy weight is key to a healthy recommendation for weight gain during pregnancy and for assessment of pregnancy risks.

Medical professionals typically calculate pre-pregnancy weight from the woman’s self-reported pre-pregnancy weight, because actual pre-pregnancy weight is often unknown. Research indicates high correlation between self-reported and actual pre-pregnancy weights in the US adult population^{3,4} and in pregnant adolescents⁵. However, in this study of Latina women, who were recent immigrants, the self-reported pre-pregnancy weights were not reliable and led us to work on an imputation method.


2)_Description of Study and Participants

The study in which pre-pregnancy weights were imputed was called “Healthy Mothers on the Move (2003-2006)”. The participants were 276 pregnant Latina women in southwest Detroit. The goal of the study was to reduce the risk factors for gestational and type 2 diabetes, through a culturally tailored education program on diet, exercise, and healthy pregnancy. Of the 276 women, 190 had pre-natal care data.

For an imputation model, the above dataset was combined with an earlier study on 1,041 Latina women from the same community, known as the “Maternal Health & Pregnancy Outcomes Among Hispanics Study (1998 - 2002)”. In this second study, all of the women had pre-natal care data.

The figures below display estimates of weight gained and loss during the first trimester among Healthy Mothers on the Move, based on self-reported pre-pregnancy weights. Reasonable values are a few pounds.

Figures 1a, b: Weight Trajectories Based on Self-Reported Pre-Pregnancy Weight





3)_Method for Imputing Missing Pre-Pregnancy Weights

The imputation dataset combined the 276 women from “Healthy Mothers on the Move” and the earlier study with 1,041 women, along with an indicator for the dataset. Pregnancy weight gain is small in the first trimester, 1.1 - 4.4 pounds, and approximately linear in the second and third trimesters. First, a dataset with complete covariate values was needed and multiple imputation with Proc MI was used to impute the missing covariates. Second, a linear mixed model (LMM), with a random intercept and slope, was developed for estimation of gestational weight by gestational week. The random intercept and slope made person-level estimates possible. In this section, the focus will be on imputing missing covariates.

In an imputation model, all variables associated with the outcome or with the probability of missingness should be included. To check whether a variable is significantly associated with missing, create missing data indicators. Then, compare the outcome by whether the outcome was missing.

For example, let the outcome = gestational weight at week 13, the beginning point of linear weight gain.

```
MissWt13 = Missing(wt13); /* 1 if missing, 0 if present */
```

```
/* Compare categorical variables with Pearson Chi-Square or Fisher Exact test in Proc Freq */
```

```
Proc Freq Data=HealthyMoms;
Tables (parity workforpay) * MissWt13 / ChiSq Exact;
Run;
```

```
/* Compare numerical variables with T-test or Wilcoxon test */
```

```
Proc TTest Data=HealthyMoms;
Class MissWt13;
Var ageyrs hip waist upperarm triceps;
```

```
Run;

Proc NPar1Way Wilcoxon Data=HealthyMoms;
Class MissWt13;
Var ageyrs hip waist upperarm triceps;
run;
```

The following variables were included in the imputation model:

- Healthy MOMS participant (y/n)
- Randomization status
- Age
- Parity (# live births)
- Height
- Number of people living in house
- Working for pay (y/n)
- # Years lived in the United States
- Anthropometry in the second trimester (waist, hip, upper arm circumference; triceps skinfold)
- Measured weights (weeks 13-27)

/* SAS Program to impute missing data */

```
proc mi data = MOMSand1000 nimpute = 5 out = OUTMI seed = 451
  min =      0 0 18 0 . 1 0
            0 14 . . .
            9 80 80 80 80 80 80 80 80 80
            80 80 80 80 80 80
  max =      1 1 . . . . 1
            . 28 . . .
            . . . . . . . . .
            . . . . . ;
  em maxiter = 500;          /*added for MLE to converge*/
  mcmc initial=em(maxiter=500); /*added for posterior mode to converge*/
  var  MOMS randomization ageyrs parity htmtrs pplinhouse workforpay
      YearUS anthrowk_tri2 hip_tri2 waist_tri2 upperarm_tri2
  triceps_tri2 wt27 wt23 wt26 wt17 wt24 wt25 wt18 wt20 wt19 wt15 wt21 wt16 wt22
  wt14 wt13;
run;
```

The “em” statement tells SAS to get the initial estimate by using maximum likelihood estimation and estimating the parameters by the Expectation-Maximization (EM) algorithm. The “min” and “max” statements prevent Proc MI from generating impossible values.

Five imputations were chosen because the Proc MI imputation table indicated that m = 5 imputations correspond to over 95% relative efficiency^{6,7}.

m = Number of imputations

λ = Proportion of observations with incomplete data.

RE = Relative Efficiency = Ratio between variance of estimates based on m imputations to the asymptotic variance from infinite imputations.

$$RE = (1 + \lambda/m)^{-1}$$

Table 1: Number of Imputations and Relative Efficiencies (Proc MI Documentation)

Relative Efficiencies λ					
m	10%	20%	30%	50%	70%
3	0.9677	0.9375	0.9091	0.8571	0.8108
5	0.9804	0.9615	0.9434	0.9091	0.8772
10	0.9901	0.9804	0.9709	0.9524	0.9346
20	0.9950	0.9901	0.9852	0.9756	0.9662

In the Proc MI output, check out the variance table and make sure that the relative efficiency of the estimates is at least 95%.

Table 2: Proc MI Relative Efficiency Information

Variable	Variance				Relative	Fraction	Relative Efficiency
	Between	Within	Total	DF	Increase	Missing	
					in	Information	
htmtrs	2.35E-08	2.95E-06	2.98E-06	1174.7	0.0096	0.0095	0.9981
pplnhouse	8.28E-06	0.003257	0.003267	1210.9	0.0031	0.0030	0.9994
workforpay	5.26E-07	0.000109	0.00011	1199.1	0.0058	0.0058	0.9989
YearUS	0.000755	0.020638	0.021544	769.73	0.0439	0.0429	0.9915
anthrowk_tri2	0.001131	0.012552	0.013909	303.94	0.1081	0.1018	0.9800
hip_tri2	0.001806	0.085774	0.087942	1006.4	0.0253	0.0249	0.9950
waist_tri2	0.000413	0.103042	0.103537	1203.8	0.0048	0.0048	0.9990
upperarm_tri2	7.56E-05	0.012281	0.012371	1189.8	0.0074	0.0074	0.9985
triceps_tri2	0.002065	0.047929	0.050407	681.25	0.0517	0.0503	0.9900

4) Linear Mixed Models

The SAS Proc Mixed code implements the following random effects:

- Let i denote the i th participant.
- Let t = gestational week.
- Let β 's be the fixed effects (i.e., population-average estimates).
- Let b 's be the random effects (i.e., participant-specific estimates).

$$\text{Weight}_{it} = \beta_0 + \beta_1 * t + b_{0i} + b_{1i} * t + (\beta\text{'s} \times \text{other covariates}) + \epsilon_{it}$$

/* Subject-specific slope and intercept */

/* The Outpred dataset estimates $\text{Weight}_{it} = \beta_0 + \beta_1 * t + b_{0i} + b_{1i} * t + (\beta\text{'s} \times \text{other covariates})$ for each participant */

```
proc mixed data = topredictwt method = REML NOCLPRINT;
model wt = gestwk parity YearUS height triceps anthrowk / outpred = pdat
solution ddfm=kr;
random int gestwk / subject = id type = un g gcorr;
by _imputation_;
run;
```

/* SAS Code for Population Average Slope, Based on SolutionF*/

```
proc mixed data = slope method = REML NOCLPRINT;  
model wt = gestwk parity YearUS height triceps anthrowk / solution ddfm=kr;  
random int gestwk / subject = id type = un g gcorr;  
by _imputation_;  
ods output solutionF=MixEst;  
run;
```

```
proc mianalyze data=MixEst;  
by effect;  
modeleffects Estimate;  
stderr standarderr;  
where standarderr ~=. ;  
Ods output parameterestimates = mianalyzeout varianceinfo= varianceinfo;  
Run;
```

/* How estimates are combined in Proc MI Analyze⁷ */

For m imputations, the point estimate is the mean of the m imputations.

The variance is a combination of the within-imputation variance and the between-imputation variance.

- Let r = number of variables in the imputation model.
- Let $\hat{\theta}_i$ be the $r \times 1$ vector of parameter estimates from the i th imputation.
- $\bar{\theta} = \frac{1}{m} \sum_{i=1}^m \hat{\theta}_i$ = Combined parameter estimates from the m imputations.
- W_i be a $r \times 1$ vector of within-imputation variances for the i th imputation.
- $\bar{W} = \frac{1}{m} \sum_{i=1}^m W_i$ = Within-Imputation Variance.
- $B = \frac{1}{(m-1)} \sum_{i=1}^m (\hat{\theta}_i - \bar{\theta})^2$ = Between-Imputation Variance, dimension $r \times 1$.
- $T = \bar{W} + \left(1 + \frac{1}{m}\right) B$ = Total Variance of parameter estimates, dimension $r \times 1$.

Estimated Weight at Week 13 by Using Self-Reported Weight. In contrast to the linear mixed models, weight at week 13 was estimated by adding 3.3 pounds to the self-reported pre-pregnancy weight. The 3.3 for estimated first trimester weight gain was obtained from a publication by Yu and Nagey⁸.


5)_ Graphical Comparison of Variances (Histograms, Box Plots, Limits of Agreement).

For a subset of women with both self-reported pre-pregnancy weight (SRPW) and a measured weight at week 13 ($N = 138$), we compared the measured weight at week 13 with:

- Prediction results calculated using a fixed slope = 1.07 pounds/week.
- Prediction results calculated by SAS using a subject-specific slope added to 1.07 pounds/week.
- Estimated weight at gestational week 13 (EFTW) = SRPW + 3.3 lbs

Graphs was made to compare the variance between the estimated and actual weights at week 13 between the three methods. Note the overlap between the population average intercept and slope with the subject-specific intercept and slope.

Figure 2: Histograms of Error Between Estimated and Actual Weights at Week 13




```

/* SAS code to produce overlapping histograms */
proc sgplot data = histogram;
  histogram actual_SR / showbins;
  histogram meas_fix / showbins;
  histogram meas_RE / showbins;
run;

```

Figure 3: Box Plots of Error Between Estimated and Actual Weights at Week 13



```

/* SAS code to produce box plots */
/* Combine 3 separate estimate columns into single column with group variable*/
data newdata1;  set histogram (keep = id actual_SR);
  rename actual_SR = difference;
  group = 1; run;

data newdata2;  set histogram (keep = id meas_fix);
  rename meas_fix = difference;
  group = 2; run;

data newdata3;  set histogram (keep = id meas_RE);
  rename meas_RE = difference;
  group = 3; run;

/* Label group variable */
proc format;
  value estfmt 1="Measured - SRP"
    2="Measured - fixed slope"
    3="Measured - subject-specific slope";
run;

data combined;
  merge newdata1-newdata3;
  by id;
  format group estfmt.;
  label difference='Weight Difference';
run;


/* Box Plot */
proc sgplot data = combined;
  hbox difference / category = group;
run;

```


Next, consider the “Bland-Altman Plot⁹”, which displays the means for two estimation methods, along with scattered data. The limits of agreement are defined as mean \pm 1.96 X standard deviation. The Bland-Altman plot below compares the population-average slope and self-reported methods. This plot is surprisingly easy in SAS Proc SGPanel by using the refline feature. “Upper” and “Lower” were columns added to the dataset and set equal to mean \pm 1.96 X standard deviation.

The graphs were annotated to include the Pearson correlation coefficients. Although both methods had high correlation, the model-based average slope had a smaller variance.

Figure 4: Bland-Altman Plot Weight with Limits of Agreement



```

/*Make a high-resolution graphic for the weight estimation methods */
ods html path='.' (url=None) image_dpi=600 style=journal;
ods graphics / imagefmt=jpeg height=3in width=4in border=off imagename='Limits
of Agreement LBS';
title "Estimation Method";
proc sgpanel data = panelplot;
  panelby type / rows = 1 columns = 2 novarname;
  scatter x = wt y = diff;
  refline 0 / lineattrs = (color = black);
  refline upper / lineattrs = (pattern = 2 color = black);
  refline lower / lineattrs = (pattern = 2 color = black);
run;
quit;
ods html close;

```

The table below of predicted weights further illustrates the information in the graphs. Using the population-average parameters is close in accuracy to using the subject-specific intercept and slope. Both methods substantially reduce the variance, compared to adding a constant to self-reported pre-pregnancy weight.

Table 3: Differences Between Measured and Estimated Weights at Week 13

		<i>Predicted weights</i>	
	<i>EFTW* based on SRPW</i>	Population average slope	Subject-specific slope
Mean	-0.2lbs	0.8 lbs	0.4 lbs
Minimum	-83.1 lbs	-6.6 lbs	-6.3 lbs
Maximum	31.7 lbs	11.3 lbs	6.9 lbs
≥5 lbs	77 (55.8%)	9 (6.5%)	5 (3.6%)
<1 lb	13 (9.4%)	41 (29.7%)	60 (42.0%)

* EFTW = Estimated weight at gestational week 13.

6) Testing for Variance Differences in Proc Mixed with a Random Intercept.

To compare variances between independent groups, an F-test would be used for two groups or Levene's test for more than groups. However, these tests do not apply when comparing estimation methods because the groups are not independent.

To compare the variances between estimation methods, use a linear mixed model with either: a random intercept for the categorical variable, Group, where Group = estimation method OR the repeated statement with Group = estimation method. Use of the random or repeated statements with the Group= option will compute separate variance for each group.

Let i = index for woman.

Let j = 1 for self-reported weight, 2 for population-average slope weight.

$$\text{Difference}_{ij} = \beta_0 + \beta_1 \text{Group}_{ij} + b_{0i} + \varepsilon_{ij}$$

/* SAS code to compare variances between estimation methods */

```
proc mixed Data=ForFTest Method=REML;
Class ID EstMethod;
Model Diff= EstMethod /Solution;
```

***** Repeated OR Random statement with Group=estimation method. ***;**

```
Repeated/Group= EstMethod Subject=ID;
```

/* Repeated statement allows for different variances by group */

```
Random Int/Subject=ID Group=EstMethod; /* 2 observations for each ID, self-
reported & predicted */
```

```
Run;
```

```
ods graphics off; ods html close;
```

```
proc mixed Data=ForFTest Method=REML;
```

```
Class ID Group;
```

```
Model Diff=Group/Solution;
```

```
Repeated/Subject=ID;
```

```
Run;
```

/* Likelihood Ratio Test for Separate Variances by Group */

```
Proc IML;
```

```
LRT_Full=1067.0;
LRT_Reduced=1236.7;
LRT=abs(LRT_Full-LRT_Reduced);
p_value=.5*(1-probchi(LRT,1)); /* . */
print LRT[format=7.2] p_value[format=7.4];
quit;
```

7)_Comparing Classification in Proc Freq with the McNemar Statistic and Cohen's Kappa.

For validation of the weight-at-week 13 prediction, we used a dataset from another study, in which 125 women had known pregnancy weights at week 13. An indicator for excessive weight gain was calculated, based on the American College of Obstetrics and Gynecology criteria¹. Excessive weight gain was calculated by using predicted weight and by using self-reported weight. Both predicted and self-reported excessive weight gain categories were compared to actual excessive weight gain categories. Because independent populations are not being compared, the traditional Pearson chi-square test is not appropriate.

McNemar's chi-square¹⁰ is used to compare matched pairs of categories. In this case, we are comparing the error rates between actual and self-reported to actual and predicted. The error rates between predicted (2.94%) and self-reported (15.69%), compared with actual, are significantly different, based on the McNemar test, $p = .0008$.

For a 2 x 2 table, the McNemar chi-square, Q_M , is computed as follows and is based on off-diagonal elements.

n_{11}	n_{12}
n_{21}	n_{22}

$$Q_M = (n_{12} - n_{21})^2 / (n_{12} + n_{21}).$$

```
/* SAS Code to compare error rates with the McNemar Test */
Data McNemarWtGain125;
set Analysis (keep=id ExcessPregGain_A ExcessPregGain_P ExcessPregGain_S);
if (ExcessPregGain_S=ExcessPregGain_A) then SR_Accurate=1;
if (ExcessPregGain_S NE ExcessPregGain_A) then SR_Accurate=0;

if (ExcessPregGain_P=ExcessPregGain_A) then PR_Accurate=1;
if (ExcessPregGain_P NE ExcessPregGain_A) then PR_Accurate=0;
run;

/* Agree option on the Proc Freq table statement will display McNemar test */
proc freq data=McNemarWtGain125;
tables SR_Accurate*PR_Accurate/agree;
run;
quit;
```

Cohen's kappa measures agreement between two observations on the same scale¹¹ and is calculated in the following way for k X k table with a total of N elements. $\kappa = 1 - (1 - p_o)/(1 - p_e)$.

Let p_o = observed probability of agreement between two observations.

$$p_o = (\text{sum of diagonal elements})/N = \sum_{i=1}^k \frac{n_{ii}}{N}.$$

$$\text{Let } p_e = \text{probability of chance agreement} = (\text{sum of each row})(\text{sum of each column})/N^2 = \frac{1}{N^2} \sum_{i=1}^k n_{i+} n_{+i}$$

For the 2 X 2 table above, $p_o = (n_{11} + n_{22})/N$ and $p_e = [(n_{11} + n_{12})(n_{11} + n_{21}) + (n_{21} + n_{22})(n_{12} + n_{22})]/N^2$.

The Cohen's kappa statistics differed significantly for predicted and actual, (95% CI = 0.94 (0.87, 1.00)), compared to self-reported and actual, (95% CI = 0.68 (0.53, 0.82)). The non-overlap between the confidence intervals for kappa indicates that predicted excessive weight gain has a significantly higher level of agreement with actual excessive weight gain. The SAS output for kappa includes the 95% confidence interval.

```
/* SAS code for Cohen's kappa. Add agree option to Tables statement */
proc freq data = analysis;
tables ExcessPregGain_A ExcessPregGain_P ExcessPregGain_S;
tables ExcessPregGain_S*ExcessPregGain_A / agree out=Gain_scat_acat;
tables ExcessPregGain_P*ExcessPregGain_A / agree out=Gain_pcat_acat;
run;
```

8) Logistic Regression. Comparison of Areas Under ROC Curves Between Models

Another way of comparing predicted and self-reported pregnancy weights is by comparing the ROC curves between logistic regression models with predicted and self-reported weights as covariates. The difference in areas under two ROC curves has a chi-square distribution¹² and can be evaluated with SAS Proc Logistic.

In the example below, models for “low birth weight” (LBW2) is being compared between models with self-reported and predicted pregnancy weights. To compare two ROC curves in SAS, write out the model statement to list all variables under consideration and be sure to include the “NOFIT” option. Then, list the actual variables in the two models on the ROC statements. To compare the models, use the ROCContrast statement. The results of this analysis using pregnancy weights produced a p-value of .0506, marginal significance.

```
*** Compare ROC curves ***;
PROC LOGISTIC DATA=analysis2 plots=roc(id=prob);
  model LBW2 = age_cen bmi_sr bmi_pwl3 gain28s gain28p par_cat scrn_cen devwks/
nofit firth;
  ROC 'bmi_pwl3' age_cen bmi_pwl3 gain28p par_cat scrn_cen devwks;
  ROC 'bmi_sr' age_cen bmi_sr gain28s par_cat scrn_cen devwks;
  ROCCONTRAST / ESTIMATE=ALLPAIRS E;
RUN;
```

9) Conclusions.

Estimating pregnancy weights, by using multiple imputation and a random effects model, were more accurate than self-reported weights, indicated by:

- Variance comparison with a linear mixed by using SAS Proc Mixed.
- Graphically displaying histograms, box plots, and Bland-Altman plot via SAS Graphics.
- Categorical comparisons, McNemar test and Cohen's kappa, from SAS Proc Freq.
- Comparing areas under the ROC curves by using SAS Proc Logistic.

For future health studies with missing data, a possible solution is to impute missing covariates by combining the current study data with previous data from the same population. The next step would be produce individual-level estimates with a random effects model. To validate the results, compare the predicted values to actual values with a validation dataset. A wide range of graphics and statistical resources are available in SAS software.

10) Acknowledgements.

- Study participants and Women's Health Advocates (Community Health Workers).
- Community Health and Social Services (CHASS) of Detroit, MI.
- Funding: NIH/NIDDK grants #R18 DK062344, #R40 MC00115-03.
- Michigan Diabetes Research and Training Center grant #DK020572.

REFERENCES

1. Obstetricians and Gynecologist, American College of. Weight gain during pregnancy. *Women's Healthcare Physicians*. 2013(548):1-2.
2. Clinic TM. Pregnancy weight gain: What's healthy?
<http://www.mayoclinic.org/healthy-lifestyle/pregnancy-week-by-week/in-depth/pregnancy-weight-gain/art-20044360>. Accessed 6/16, 2017.
3. Stunkard AJ, Albaum JM. **The accuracy of self-reported weights.** *Am J Clin Nutr*. 1981;34(8):1593-1999.
4. Stewart AW, Jackson RT, Ford MA, Beaglehole R. Under estimation of relative weight by use of Self reported height and weight. *Am JEpidemiol*. 1982;125(1):122-126.
5. Stevens-Simon C, Nakashima II, Andrews D. Weight gain attitudes among pregnant adolescents. *Journal of Adolescent Health*. 1993;14(5):369-372.
6. Institute S. Multiple imputation efficiency.
http://documentation.sas.com/?cdclid=statcdc&cdcVersion=14.2&docsetId=statug&docs etTarget=statug_mi_details54.htm&locale=en. Accessed 8/2, 2017.

7. Little RJA, Rubin DB. *Statistical analysis with missing data*. 2nd ed. New York: John Wiley; 2002.
8. Yu SM, Nagey DA. Validity of self-reported pregravid weight. *Annals of Epidemiology*. 1992;2(5):715-721.
9. Bland JM, Altman DG. Statistical methods for assessing agreement between two methods of clinical measurement. *Lancet*. 1986;1(8476):307-310.
10. McNemar Q. Note on the sampling error of the difference between correlated proportions or percentages. *Psychometrika*. 1947;12:153-157.
11. Cohen J. A coefficient of agreement for nominal scales . *Educational and Psychological Measurement*. 1960;20:37-46.
12. DeLong ER, DeLong DM, Clarke-Pearson DL. Comparing the areas under two or more correlated receiver operating curves: A nonparametric approach. *Biometrics*. 1988;44:837-845.

CONTACT INFORMATION

Your comments and questions are welcome.

Brandy R. Sinco, Statistician and Programmer/Analyst
University of Michigan School of Social Work
1080 S. University St.
Box 183
Ann Arbor, MI 48109-1106

Phone: 734-763-7784

E-Mail: brsinco@umich.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.